


SUPPLEMENTAL MATERIAL


Supplementary Material 1 – Heterogeneity of individual kinetic profiles

Figure S1a – Spaghetti plots




All 223 individual profiles were plotted for PSA and CTC. Each color corresponds to one individual.

Figure S1b - Different types of observed individual kinetic profiles



Blue curves represent the PSA kinetics and red curves the CTC kinetics. Vertical dashed lines represent the treatment cycles: chemotherapy in black, hormonotherapy in green, and both treatments administered simultaneously in purple.

Figure S1c - Different types of simulated individual kinetic profiles




Blue curves represent PSA kinetics, red curves CTC kinetics, and black curves the latent variable kinetics. Vertical dashed lines represent the treatment cycles.

Supplementary Material 2 – Model including the delayed compartments

To estimate the CTC lifespan, the model was implemented in NONMEM using delayed compartments with the ALAG function. Figure S2 represents the model including the delayed compartments.

Figure S2 – Model including the delayed compartments



This model was described by the following equations:

$$\left\{ \begin{array}{l}
\frac{dA_c}{dt} = -K_c \times A_c \\
\frac{dA_c D}{dt} = -K_c \times A_c D \\
\frac{dA_h}{dt} = -K_h \times A_h \\
\frac{dA_h D}{dt} = -K_h \times A_h D \\
\frac{dLV}{dt} = Kin_{LV} \times \left(1 - \frac{A_c}{A50_c + A_c}\right) \times \left(1 - \frac{A_h}{A50_h + A_h}\right) - Kout_{LV} \times LV \\
\frac{dLVD}{dt} = Kin_{LV} \times \left(1 - \frac{A_c D}{A50_c + A_c D}\right) \times \left(1 - \frac{A_h D}{A50_h + A_h D}\right) - Kout_{LV} \times LVD \quad (if t > LS) \\
\frac{dLVD}{dt} = 0 \quad (if 0 < t < LS) \\
\frac{dPSA}{dt} = Kin_{PSA} \times LV - Kout_{PSA} \times PSA \\
\frac{dCTC_{Total}}{dt} = K0 \times LV - K0 \times LVD
\end{array} \right.$$

A_c and A_h represent drug amounts in the chemotherapy and the hormonotherapy compartments (*Arbitrary Unit, AU*), respectively. $A_c D$ and $A_h D$ represent drug amounts in the delayed chemotherapy and the delayed hormonotherapy compartments (*Arbitrary Unit, AU*), respectively. K_c and K_h are the chemotherapy and hormonotherapy kinetics rate constants, respectively (day^{-1}). $A50_c$ and $A50_h$ respectively are the amounts of each treatment producing 50% of the maximum effect (*AU*). LV corresponds to the latent variable (*AU*), and Kin_{LV} and $Kout_{LV}$ are their production and elimination rates ($AU \cdot day^{-1}$ and day^{-1}), respectively. LVD corresponds to the delayed latent variable (*AU*). LVD is supposed to be constant and equal to $LVD0$ for times before the lifespan. Kin_{PSA} and $Kout_{PSA}$ correspond to the PSA production and elimination rates ($ng \cdot ml^{-1} \cdot day^{-1} \cdot AU^{-1}$ and day^{-1}), respectively. $K0$ is the CTC production rate ($CTC \cdot day^{-1} \cdot AU^{-1}$).

The initial conditions of the model at time 0 were as follows:

$$\left\{ \begin{array}{l} A_c(0) = 0 \\ A_cD(0) = 0 \\ A_h(0) = 0 \\ A_hD(0) = 0 \\ LV(0) = LV_0 \text{ with } LV_0 < \frac{Kin_{LV}}{Kout_{LV}} \\ LVD(0) = LV_0 \\ PSA(0) = PSA_0 \\ CTC_{Total}(0) = K_0 \times LS \times LV_0 \end{array} \right.$$

$$LS = ALAG_{A_cD} = ALAG_{A_hD} = ALAG_{LVD}$$

LV_0 and PSA_0 are the initial latent variable value (AU) and the initial PSA concentration (ng/ml), respectively. LS correspond to the CTC lifespan (day).

Supplementary Material 3 – Data file and NONMEM code

Fragment of data file:

ID	TIME	AMT	DV	CMT	MDV
1	0	1	.	1	1
1	0	1	.	2	1
1	0	1	.	5	1
1	0	1	.	6	1
1	0	1	.	4	1
1	0	.	11	4	0
1	0	.	4.127134	8	0
1	21	1	.	1	1
1	21	1	.	2	1
1	21	1	.	5	1
1	21	1	.	6	1
1	21	.	0	4	0
1	21	.	3.688879	8	0
1	42	1	.	1	1
1	42	1	.	2	1
1	42	1	.	5	1
1	42	1	.	6	1
1	42	.	0	4	0
1	42	.	3.391147	8	0

CMT: 1=Chemotherapy; 2=Hormonotherapy; 4=CTC; 5=Delayed Chemotherapy; 6=Delayed Hormonotherapy; 8=PSA.

NONMEM Code:

\$PROBLEM Joint model of PSA and CTC count kinetics in mCRPC patients

\$INPUT ID TIME AMT DV CMT MDV

\$DATA ...

\$SUBROUTINE ADVAN13 TOL=9

\$MODEL NCOMP=8

COMP = (A1) ; PK chemotherapy

COMP = (A2) ; PK hormonotherapy

COMP = (A3) ; Latent Variable

COMP = (A4) ; CTC

COMP = (A5) ; Delayed PK chemotherapy

COMP = (A6) ; Delayed PK hormonotherapy

COMP = (A7) ; Delayed Latent Variable

COMP = (A8) ; PSA

\$PK CALLFL=-2 ; Call the PK subroutine with every event record, with additional and lagged doses

MU_1=LOG(THETA(1))

LV0=EXP(MU_1+ETA(1)) ; Initial Latent Variable

MU_2=LOG(THETA(2))

Kc=EXP(MU_2+ETA(2)) ; Chemotherapy kinetic rate constant

MU_3=LOG(THETA(3))
 Kh=EXP(MU_3+ETA(3)) ; Hormonotherapy kinetic rate constant

MU_4=LOG(THETA(4))
 A50c=EXP(MU_4+ETA(4)) ; Amount of chemotherapy producing 50% of the maximum effect

MU_5=LOG(THETA(5))
 A50h=EXP(MU_5+ETA(5)) ; Amount of hormonotherapy producing 50% of the maximum effect

MU_6=LOG(THETA(6))
 KOUTLV=EXP(MU_6+ETA(6)) ; Latent Variable elimination rate constant

MU_7=THETA(7)
 SFLV=EXP(MU_7+ETA(7))
 KINLV=(TS0*KOUTTS)/((SFLV)/(1+(SFLV))) ; Latent Variable production rate

MU_8=LOG(THETA(8))
 KINPSA=EXP(MU_8+ETA(8)) ; PSA production rate

MU_9=LOG(THETA(9))
 KOUTPSA=EXP(MU_9+ETA(9)) ; PSA elimination rate constant

MU_10=LOG(THETA(10))
 PSA0=EXP(MU_10+ETA(10)) ; Initial PSA concentration

MU_11=THETA(11)
 K0=MU_11+ETA(11) ; CTC production rate

MU_12=THETA(12)
 ALAG5=MU_12+ETA(12) ; Delay duration, lifespan (LS)
 ALAG6=ALAG5
 ALAG7=ALAG5

F4=K0*ALAG5 ; Initial condition for CTC: CTC(0)=K0*LS*LV0 (LV0=1)

MU_13=LOG(THETA(13))
 OVDP=EXP(MU_13+ETA(13)) ; Overdispersion

MU_14=LOG(THETA(14))
 W1=EXP(MU_14+ETA(14)) ; Standard deviation for PSA residual error

; Initial Conditions at time 0:

A_0(1)=0
 A_0(2)=0
 A_0(3)=LV0
 A_0(5)=0
 A_0(6)=0
 A_0(7)=LV0
 A_0(8)=PSA0

\$DES

DADT(1)=-Kc*A(1) ; Time course of chemotherapy amount
 DADT(2)=-Kh*A(2) ; Time course of hormonotherapy amount
 DADT(3)=KINLV*(1-(A(1)/(A50c+A(1))))*(1-(A(2)/(A50h+A(2))))-KOUTLV*A(3) ; Time course of LV
 DADT(5)=-Kc*A(5) ; Delayed time course of chemotherapy amount
 DADT(6)=-Kh*A(6) ; Delayed time course of hormonotherapy amount

DADT(7)=KINLV*(1-(A(5)/(A50c+A(5))))*(1-(A(6)/(A50h+A(6))))-KOUTLV*A(7) ; Delayed time course of LV

A7=LV0
IF(T.GT.ALAG5) A7=A(7)

DADT(4)=K0*A(3)-K0*A(7) ; Time course of total CTCs
DADT(8)=KINPSA*A(3)-KOUTPSA*A(8) ; Time course of PSA

\$ERROR

LAMB=A(4)*0.0015 ; Expected number of CTCs in the aliquot

PSA=A(8)
nCTC=DV

; Factorial:
LFAC=GAMLN(nCTC+1.)

;gamma functions of the negative binomial model expression:

LGAM1=GAMLN(nCTC+1/OVDP)
LGAM2=GAMLN(1/OVDP)
LTRM1=(LOG(1/(1+OVDP*LAMB)))*(1/OVDP)
LTRM2=(LOG(LAMB/(LAMB+1/OVDP)))*(nCTC)

;Logarithm of the Negative Binomial distribution:

LNB=LGAM1-LFAC-LGAM2+LTRM1+LTRM2

; CTC observation:

IF (CMT.EQ.4) THEN
F_FLAG=2
Y=-2*LNB ; -2 Log Likelihood
ENDIF

; PSA observation:

IF (CMT.EQ.8) THEN
F_FLAG=0
IPRED=LOG(PSA)
Y=IPRED+W1*ERR(1)
IRES=DV-IPRED
IWRES=IRES/W1
ENDIF

\$THETA

1 FIX ; LV0
(0,0.5) ; Kc
(0,0.5) ; Kh
(0,0.0001) ; A50c
(0,0.006) ; A50h
(0,0.004) ; KOUTLV
(6) ; SFLV
(0,3) ; KINPSA
(0,0.008) ; KOUTPSA
(0,150) ; PSA0
(0,300) ; K0
(0,50) ; ALAG5
(0,3) ; OVDP
(0,0.3) ; W1

\$OMEGA

0 FIX ; LV0

\$OMEGA BLOCK(9)

0.5 ; Kc
0.01 0.5 ; Kh
0.01 0.01 0.5 ; A50c
0.01 0.01 0.01 0.5 ; A50h
0.01 0.01 0.01 0.01 0.5 ; KOUTLV
0.01 0.01 0.01 0.01 0.01 0.5 ; SFLV
0.01 0.01 0.01 0.01 0.01 0.01 0.5 ; KINPSA
0.01 0.01 0.01 0.01 0.01 0.01 0.01 0.5 ; KOUTPSA
0.01 0.01 0.01 0.01 0.01 0.01 0.01 0.01 0.5 ; PSA0

\$OMEGA BLOCK(2)

10 ; K0
1 10 ; ALAG5

\$OMEGA

0.5 ; OVDP
0 FIX ; W1


\$SIGMA

1 FIX

\$ESTIMATION METHOD=SAEM LAPLACE INTER NUMERICAL

Supplementary Material 4 – Categorical VPCs for the dichotomized CTC count (<5 vs ≥ 5)

Figure S3 – Categorical VPCs (CTC count <5 vs CTC count ≥ 5)



The probabilities of having a number of CTCs lower than 5 or greater than 5 were plotted versus time. Red areas are the 95% confidence intervals of the simulated median probabilities. Blue lines are the observed probabilities.